SNOOP:

A COMPUTER PROGRAM

For 2- and 3-Dimensional Plotting

by Warren E. Frayer

U.S. FOREST SERVICE RESEARCH PAPER NE-91
1968

NORTHEASTERN FOREST EXPERIMENT STATION, UPPER DARBY, PA.
FOREST SERVICE, U. S. DEPARTMENT OF AGRICULTURE
RICHARD D. LANE, DIRECTOR

The Author . . .

WARREN E. FRAYER received a Bachelor of Science degree in forestry at The Pennsylvania State University in 1961. He earned a Master of Forestry degree at Yale University in 1962 and a Doctor of Forestry degree from the same institution in 1965. He has worked as a research forester on the Northeastern Forest Experiment Station's Forest Survey Project, and was a mathematical statistician in the Station's biometrics section when this paper was prepared. He is now on the faculty of Colorado State University at Ft. Collins, Colorado.

Contents

THE NEED FOR SNOOPING

COMPUTER programs for two-dimensional (2D) plotting are numerous, and at least one is available in the library of most computer installations. However, in the process of model-building, researchers often want to examine graphed data for the presence of possible interactions of independent variables. This type of data "snooping" is common; but, to make use of available (canned) program decks, researchers must often resort to separating data into small sets and constructing and comparing several two-dimensional graphs.

To help with this chore, we have developed a computer program called SNOOP.

WHAT SNOOP WILL DO

Program SNOOP allows examination of interactions by extending two-dimensional (2D) graphing procedures to three-dimensional (3D) plotting. A variable hereafter referred to as a plotting character (PC) denotes the third dimension. The PC can assume values from 0 to 9 and is designated by the content of a selected card column in the observation data. The numeric characters that appear on the completed graph are the actual PC values that appear on the cards (a blank results in a PC value of 0).

For example, if volume, diameter, and height are measured on a group of felled trees, the 2D relationship of volume and diameter can be plotted on the same graph and up to 10 values (or classes) of the plotting character (height) can be distinguished. Examination of the graph may then result in selections of appropriate transformations, interaction terms, and/or weights to be applied in regression analysis. This procedure is not necessary for so simple a problem as postulated here, but extension of the possible uses of 3D plotting should be obvious. However, because of its simplicity, a volume-estimation problem is used as an example later.

Two-dimensional plotting is also available to the user of SNOOP. In this case the numeric characters that appear on the completed graph are frequencies of occurrence. The resulting character at any set of coordinates may then be blank (no occurrence), 1 to 9, or 0 (10 or more).

If overplots are attempted, messages (containing the coordinates and, in the case of 3D plotting, the PC values) are printed after the appropriate graph. An attempted overplot would occur when 2 or more observations have identical coordinates for 3D plotting, or 11 or more observations for 2D plotting.

When several dependent (Y) and independent (X) variables are included in one job, a set of fully labelled graphs is made, including each Y-X combination. And, for 3D, a separate set of graphs is made for each PC specified. The program automatically scales the graphs for each variable used. Other flexibility available to the user, barring certain properties given in the section *Program Restrictions,* can be summarized as follows:

> 2D and 3D jobs can be included in the same run in any order. As many jobs as desired may be included in a single run. Observation values can be negative, zero, and positive. Order of the variables on the observation cards is completely flexible.
>
> PC's can be located anywhere on the observation cards. Multiple cards per observation can be used.

WHEN TO USE THE PROGRAM

The major use of program SNOOP is in the model-building stage of an estimation problem. A segment of data can be plotted in various ways, and the graphs can be examined for logical trends in dependent-variable values as responses to changes in the values of independent variables. Models can then be formed for regression analyses with the remaining data.

Consequently the value of the program is easily recognized when the researcher is attempting to explain previously unfamiliar functional relationships between many variables. With only a few variables, many observations, and expectations that limit the possibilities of functional-relationship alternatives, the researcher can go directly to the use of one of the canned regression routines—for example, Furnival[1]—available at any large computer center.

HOW TO USE THE PROGRAM

The program is activated by a control deck supplied by the user. The control deck consists of cards containing the observations and information designating a specific manner in which the observations are to be handled by the program. The following instructions describe the complete control-deck setup (all underlined characters *must* be punched exactly as they appear in the instructions):

Card group	*Number of cards*	*Title*	*Card columns*	*Content*
1	1	Job description	1-2	*2D*—For 2-dimensional plotting.

[1] Furnival, G. M. MORE ON THE ELUSIVE FORMULA OF BEST FIT. Soc. Amer. Foresters Proc. 1964: 201-207.

				3D—For 3-dimensional plotting.
			3-4	Number of Y's (NYS), right adjusted.
			5-6	Number of X's (NXS), right adjusted (does not include PC variables).
			7-8	Number of plotting characters (NPC), right adjusted. Use *00* (or blanks) for 2D.
			9-11	Number of observations (NOBS), right adjusted.
			12	Number of cards per observation (NCARDS). The use of *0* (or blank) will be interpreted as *1*.
			13-72	Alphameric job title to be printed above each graph.
2	1	Format	1	*1*—There is no format continuation card.
				2—The format specification is continued on the format continuation card.
			2	*(*—Left parenthesis.
			3-72	Up to 70 alphameric characters, ending with a right parenthesis, and containing a FORTRAN BCD format[2] (F-specifications), which

[2] Refer to any text on FORTRAN IV for descriptions of BCD formats (F-specifications as used here and E-specifications as used in portions of SNOOP output).

				describes the format of the Y's and X's on the observation cards. The right parenthesis is put on the format continuation card if column 1 of the format card = 2.
3	0 or 1	Format continuation	1-72	Continuation of the format specification described for columns 3-72 of the format card.
4	NYS + NXS	Variable labels	1	Y—If the variable is a dependent variable. X—If the variable is an independent variable.
			2	Blank.
			3-52	Alphameric label (identifying information) associated with the variable. This group of cards must be arranged in the order that the variables appear on the observation cards.
5	0 if 2D, NPC if 3D	3D plotting characters	1-18	*PLOTTING CHARACTER*
			19	Blank.
			20	Number of the card (of the observation set) on which the PC value is found. The use of 0 (or blank) will be interpreted as 1.
			21-22	Column number (right-adjusted) in which the

				PC value is found.
			23	Blank.
			24-31	Alphameric label (identifying information) associated with the PC.
6	NOBS xNCARDS	Observations	1-80	One or more cards for each observation as designated by the format cards.
7	1	Job control	1-8	*CONTINUE*—If more than one job is included in the run. Card groups 1-7 are repeated after the job control card. *DONE* (left adjusted) —Designates the end of the last job in the run.

An example of a control deck is shown in the appendix.

PROGRAM RESTRICTIONS

Job restrictions applying both to 2D and 3D plotting are:

$$NYS + NXS \leq 50$$
$$NYS \geq 1$$
$$NXS \geq 1$$
$$2 \leq NOBS \leq 500$$
$$1 \leq NCARDS \leq 9$$

Job restrictions pertaining to 3D plotting only are:

$$(NYS + NXS + NPC) \times NOBS \leq 10{,}000$$
$$1 \leq NPC \leq 10$$

A job restriction pertaining to 2D plotting only is:

$$(NYS + NXS) \times NOBS \leq 10{,}000$$

HOW TO USE THE RESULTS

For the researcher who is developing a regression model, the graphs produced by the program can be a valuable aid in selecting transformations, interaction terms, and variance-stabilizing weights, and in ordering of variables.

As an example, let us examine the sample control deck in Appendix I and the output in Appendix II.

The first page of output consists of summary information pertaining to the entire job. Card groups 1 to 5 of the control deck are printed as well as the values of all the variables in the first observation (the Y's precede the X's in the printout). Because the first job in this example specifies 3D plotting, the values of the plotting characters for the first observation are also included in the summary information.

Now, assume for the moment that expectations of functional relationships involving cubic-foot volume, diameter, and height are vague. The trees plotted in this example could be an initial sample of trees to be measured in the construction of a standard volume table; that is, we want to express cubic volume as a function of both diameter and height, if possible. Consequently, the control deck has been set up to plot a 3D graph of cubic-foot volume as a function of diameter, using 10-foot height class as a plotting character.

The resultant graph shows that volume increases with an increase in diameter, and there is an increase in the slope of the relationship as diameter increases.

Free-hand curves as they would be drawn on the graph for several groups of the 10-foot height classes are shown in figure 1. These can be represented in three dimensions as in figure 2.

Upon close examination — by methods described by Jensen[3] — figure 2 suggests that, for a given height, volume is a function of the square of diameter. Also, for a given diameter, volume

[3] Jensen, C. E. ALGEBRAIC DESCRIPTION OF FORMS IN SPACE. U. S. Forest Serv., Cent. States Forest Exp. Sta. Misc. Pub. 57 pp. Columbus, Ohio. 1964.



Figure 1.—Hand-fitted curves for four height-class groups.

appears to be a linear function of height. Thus, the interaction term of [(diameter)^2 x height] would be an appropriate independent variable for regression analysis. The user might also study the graph further to determine a weighting procedure to stabilize the variance of volume as [(diameter)^2 x height] varies.

Following the graph is the printed list of attempted overplots, including their coordinates and PC values. Following the overplots, the printed message CONTINUE signifies that the program is continuing to the next job in the run.

The second job is simply a demonstration of 2D graphing. The same observations were used as in the first job. Two graphs were specified by designating volume as a dependent variable



Figure 2.—Cubic-foot volume as a function of the joint distribution of tree diameter at breast height and total tree height.

and diameter and height as independent variables. The reader will note that the numbers printed in the resultant graphs are frequencies of occurrence. There are no overplots, and the printed message DONE signifies that the last job has been run.

INFORMATION FOR PROGRAMMERS

Program SNOOP was thoroughly tested and is operational on an IBM 7040 computer with the IBSYS monitor. The program is written in FORTRAN IV and should run with little or no modification on other computers that accept FORTRAN IV, have a 32K core, and have three tape drives or equivalent input/output devices. The logical tape assignments are:

Unit	*Use*
5	Control-deck input
6	Program output
9	Intermediate operations (scratch tape)

A special subroutine was necessary for backspacing logical unit 5 at the computer installation where this program was developed. A simple BACKSPACE 5 command or its equivalent is all that is necessary at other installations. The two places in the program where this must be done are marked with comments. A general flow diagram of the program is shown in figure 3. And a listing of the program is presented in appendix III.

ERROR MESSAGES

When the user makes an error in preparing the control deck, the program will print a message if the error is such that execution of the program would be illogical or impossible. The program then continues to scan the control deck for additional jobs. The error messages used by the program are:

NYS = XX
NYS MUST BE GREATER THAN 0 AND LESS THAN 50

NXS = XX
NXS MUST BE GREATER THAN 0 AND LESS THAN 50

NPC = XX
NPC MUST BE BLANK OR 0

NPC = XX
NPC MUST BE GREATER THAN 0 AND LESS THAN 11

NOBS = XXX
NOBS MUST BE GREATER THAN 1 AND LESS THAN 501



Figure 3.—SNOOP flow chart.

(NYS + NXS) X NOBS = XXXXXX
THIS MUST BE EQUAL TO OR LESS THAN 10,000

(NYS + NXS + NPC) X NOBS = XXXXXX
THIS MUST BE EQUAL TO OR LESS THAN 10,000

THE FOLLOWING CONTROL CARD IS MISPUNCHED OR OUT OF ORDER. . .
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
XXXXXXXXXXXXXXXXXXXXXX

THERE IS A VARIABLE LABEL WITH THE 3D PLOTTING CHARACTER GROUP

THERE IS A NON-NUMERIC CHARACTER IN OBSERVATION XXX FOR PLOTTING CHARACTER XX

After any of the above messages are printed, the following message is printed:
UNABLE TO PROCESS THIS JOB
THE PROGRAM IS NOW SEARCHING FOR THE NEXT JOB (IF ANY)

APPENDIX

I

CONTROL-DECK EXAMPLE

```
3D010101095 1RED PINE VOLUME TABLE CONSTRUCTION (PRELIMINARY PLOTTING)
1(5X,F3.1,5X,F4.1)
X                DIAMETER AT BREAST HEIGHT
Y                     CUBIC-FOOT VOLUME
PLOTTING CHARACTER 111 HT CLASS
      143   62   432
      145   72   513
      148   80   597
      150   46   351
      152   45   356
      153   40   317
      154   52   418
      156   68   561
      157   56   470
      159   75   640
      160   60   521
      163   73   657
      165   85   780
      168   80   770
      170   50   480
      173   60   609
      175   56   570
      177   58   617
      178   64   694
      179   72   790
      180   58   630
      181   69   766
      183   62   704
      184   74   840
      186   76   900
      188   64   788
      131   44   250
      132   56   335
      134   48   292
      136   47   290
      137   53   338
      139   56   369
      140   59   390
       50   20    17
       51   22    19
       52   26    25
       54   28    23
       56   24    26
       59   36    42
       60   30    36
       62   33    43
       63   24    33
       65   26    38
       65   36    52
       70   40    66
```

CONTINUED

```
 72  39   69
 74  42   79
 77  42   82
 79  45   94
 80  46  100
 83  45  102
 86  50  123
 90  30   84
 92  35  101
 94  36  111
 97  44  140
111  58  242
100  51  175
101  40  138
103  52  187
104  48  174
106  55  209
189  84 1030
193  70  880
198  90 1222
109  58  234
110  39  160
199  86 1141
200  60  801
 99  38  123
119  45  216
219  82 1330
220  95 1502
221  98 1634
223  74 1263
225  86 1530
119  65  315
120  54  260
126  70  377
230  60 1036
233  95 1750
130  40  235
130  42  241
240  97 1894
261  60 1360
263 100 2371
266  87 2104
268 101 2453
280 102 2763
113  40  170
115  64  293
202  72 1002
210  74  983
218  75 1223
116  43  199
CONTINUE
200102000912RED PINE DATA - EXAMPLES OF TWO-DIMENSIONAL PLOTTING)
1(5X,F3.1,1X,F3.0,1X,F4.1)
X              DIAMETER AT BREAST HEIGHT
X                    TOTAL HEIGHT
Y                  CUBIC-FOOT VOLUME
(SAME OBSERVATION DECK AS USED FOR THE FIRST JOB MUST BE INSERTED HERE)
DONE
```

II
OUTPUT FOR CONTROL DECK GIVEN IN I.

```
JOB NUMBER  1
JOB TITLE =RED PINE VOLUME TABLE CONSTRUCTION (PRELIMINARY PLOTTING)
3D PLOTTING

NUMBER OF DEPENDENT(Y) VARIABLES = 1
NUMBER OF INDEPENDENT(X) VARIABLES = 1
NUMBER OF OBSERVATIONS = 95
NUMBER OF CARDS PER OBSERVATION =1
NUMBER OF PLOTTING CHARACTERS = 1

THE CONTROL DECK FOR THIS JOB IS......

3D 1 1 1 1RED PINE VOLUME TABLE CONSTRUCTION (PRELIMINARY PLOTTING)
(15X,F3.1,5X,F4.1)
X              DIAMETER AT BREAST HEIGHT
Y                   CUBIC-FOOT VOLUME
PLOTTING CHARACTER 1(1 HT CLASS

....OBSERVATION CARDS ARE LOCATED HERE....

THE VARIABLES IN THE FIRST OBSERVATION ARE...
Y( 1)= 0.4320000E 02
X( 1)= 0.1430000E 02

THE PLOTTING CHARACTERS FOR THIS OBSERVATION ARE....
PC( 1)=6
```

```
                    JOB NUMBER   1
                    RED PINE VOLUME TABLE CONSTRUCTION (PRELIMINARY PLOTTING)
                    THREE-DIMENSIONAL GRAPH,PC( 1) IS DEFINED AS HT CLASS
```




```
NUMBER OF ATTEMPTED OVERPLOTS=  4

COORDINATES OF ATTEMPTED OVERPLOTS
Y= 0.4446E 01   X= 0.5115E 01   PC= 2
Y= 0.4446E 01   X= 0.5115E 01   PC= ?
Y= 0.4446E 01   X= 0.6265E 01   PC= 4
Y= 0.4446E 01   X= 0.5695E 01   PC= 3

CONTINUE
```

CONTINUED

OUTPUT FOR CONTROL DECK. — CONTINUED.



CONTINUED

Output For Control Deck. — Continued.



DONE

III
LISTING OF PROGRAM SNOOP.

```
      DIMENSION FMT(36),LABEL(50,50),ILABEL(50,50),CONLAB(10,2),TITLE(15
     1),IVAR(50),CHECK(5),IPC(10),XMAT(10120),PC(10),V(50),VV(50),KPC(72
     20),R(20),IGRAPH(50,100),XABS(5),IG(12)
C
C
C READ THE CONTROL DECK AND CHECK FOR ERRORS
C
C
C READ THE JOB DESCRIPTION CARD
      IJOB=0
  799 IJOB=IJOB+1
      READ(5,700)DEM,NYS,NXS,NPC,NOBS,NCARDS,(TITLE(I),I=1,15)
  700 FORMAT(A2,3I2,I3,I1,15A4)
      NC=NCARDS
      IF(NCARDS.EQ.0)NC=1
      IF(NCARDS.EQ.0)NCARDS=0
      WRITE(6,7501)IJOB
      WRITE(6,7511)(TITLE(I),I=1,15)
      WRITE(6,752)DEM
      WRITE(6,753)NYS
      WRITE(6,754)NXS
      WRITE(6,755)NOBS
      WRITE(6,10001)NC
10001 FORMAT(34H NUMBER OF CARDS PER OBSERVATION =I1)
      WRITE(6,756)NPC
      WRITE(6,757)
 7501 FORMAT(11H1JOB NUMBER13)
 7511 FORMAT(12H JOB TITLE =15A4)
  752 FORMAT(1H A2,9H PLOTTING)
  753 FORMAT(35HONUMBER OF DEPENDENT(Y) VARIABLES =I2)
  754 FORMAT(37H NUMBER OF INDEPENDENT(X) VARIABLES =I2)
  755 FORMAT(25H NUMBER OF OBSERVATIONS =I3)
  756 FORMAT(32H NUMBER OF PLOTTING CHARACTERS =I2)
  757 FORMAT(43HOTHE CONTROL DECK FOR THIS JOB IS.....    )
      DATA THREED/2H3D/
      DATA TWOD/2H2D/
      IF(DEM.NE.THREED.AND.DEM.NE.TWOD)GO TO 999
      IF(NYS.LE.0.OR.NYS.GE.50)GO TO 800
      IF(NXS.LE.0.OR.NXS.GE.50)GO TO 801
      IF(NOBS.LT.2.OR.NOBS.GT.500)GO TO 810
      IF(DEM.EQ.THREED)GO TO 702
      IF(NPC.NE.0)GO TO 802
      IF(((NYS+NXS)*NOBS)-10000)703,703,860
  702 IF(NPC.LE.0.OR.NPC.GT.10)GO TO 803
      IF(((NYS+NXS+NPC)*NOBS)-10000)703,703,861
  703 WRITE(6,704)DEM,NYS,NXS,NPC,NOBS,NCARDS,(TITLE(I),I=1,15)
  704 FORMAT(1HOA2,3I2,I3,I1,15A4)
C READ THE FORMAT CARDS
      READ(5,780)IFORM,(FMT(I),I=1,18)
  780 FORMAT(I1,17A4,A3)
      IF(IFORM.LT.1.OR.IFORM.GT.2)GO TO 999
      WRITE(6,781)IFORM,(FMT(I),I=1,18)
  781 FORMAT(1H I1,17A4,A3)
      IF(IFORM.EQ.1)GO TO 706
      READ(5,705)(FMT(I),I=19,36)
  705 FORMAT(18A4)
      WRITE(6,782)(FMT(I),I=19,36)
  782 FORMAT(1H 18A4)
C READ THE VARIABLE LABELS
  706 NVAR=NYS+NXS
```

CONTINUED

```
      IXX=0
      IYY=0
      DO 790 I=1,50
  790 IVAR(I)=0
      DO 707 I=1,NVAR
      READ(5,708)VAR,(LABEL(I,K),K=1,50)
  708 FORMAT(A1,1X,50A1)
      DATA YY/1HY/
      DATA XX/1HX/
      IF(VAR.NE.YY.AND.VAR.NE.XX)GO TO 999
      IF(VAR.EQ.XX)GO TO 709
      IYY=IYY+1
      IVAR(I)=IYY
      GO TO 710
  709 IXX=IXX+1
      IVAR(I)=IXX+NYS
  710 WRITE(6,791)VAR,(LABEL(I,K),K=1,50)
  791 FORMAT(1H A1,1X,50A1)
C     ORDER THE VARIABLE LABELS
      KKK=IVAR(I)
      DO 711 K=1,50
  711 ILABEL(KKK,K)=LABEL(I,K)
  707 CONTINUE
C READ THE PLOTTING CHARACTERS LABELS
      IF(DEM.EQ.TWOD)GO TO 751
      DO 750 I=1,NPC
      READ(5,712)(CHECK(IJ),IJ=1,5),ICARD,IPC(I),(CONLAB(I,K),K=1,2)
  712 FORMAT(A2,4A4,1X,I1,I2,1X,2A4)
      IF(CHECK(1).EQ.YY.OR.CHECK(1).EQ.XX)GO TO 830
      DATA T1,T2,T3,T4,T5/2HPL,4HOTTI,4HNG C,4HHARA,4HCTER/
      IF(CHECK(1).NE.T1.OR.CHECK(2).NE.T2.OR.CHECK(3).NE.T3.OR.CHECK(4).
     1NE.T4.OR.CHECK(5).NE.T5)GO TO 999
      IF(IPC(I).LE.0.OR.IPC(I).GT.80)GO TO 999
      WRITE(6,792)(CHECK(IJ),IJ=1,5),ICARD,IPC(I),(CONLAB(I,K),K=1,2)
  792 FORMAT(1H A2,4A4,2I2,1X,2A4)
      IC=ICARD
      IF(IC.EQ.0)IC=1
      IPC(I)=((IC-1)*80)+IPC(I)
  750 CONTINUE
  751 CONTINUE
C
C
C THE OBSERVATIONS ARE NOW PLACED IN A VECTOR CALLED XMAT
C
C
      WRITE(6,797)
  797 FORMAT(45HO....OBSERVATION CARDS ARE LOCATED HERE....     )
      DO 100 K=1,50
      V(K)=0.0
  100 VV(K)=0.0
      DO 110 K=1,10120
  110 XMAT(K)=0.0
      ICOUNT=0
C READ THE VARIABLES
  103 READ(5,FMT)(V(K),K=1,NVAR)
      ICOUNT=ICOUNT+1
      IF(DEM.NE.THREED)GO TO 13
      DO 9691 IJKL=1,NC
C THE FOLLOWING CALL STATEMENT IS USED TO BACKSPACE THE INPUT UNIT
      CALL IOEDIT(1)
```

CONTINUED

```
 9691 CONTINUE
C READ THE PLOTTING CHARACTERS
      KSTART=1
      KEND=80
      DO 9692 IJKL=1,NC
      READ(5,112)(KPC(K),K=KSTART,KEND)
      KSTART=KSTART+80
 9692 KEND=KEND+80
  112 FORMAT(80A1)
      DO 113 I=1,NPC
      KKPC=IPC(I)
      IRKPC=KPC(KKPC)
      DO 960 LL=1,11
      DATA IG/1H ,1H1,1H2,1H3,1H4,1H5,1H6,1H7,1H8,1H9,1H0,1H./
      IF(IRKPC.NE.IG(LL))GO TO 960
      PC(I)=LL-1
      IF(LL.EQ.11)PC(I)=0.
      GO TO 113
  960 CONTINUE
      GO TO 835
  113 CONTINUE
      GO TO 13
10013 WRITE(6,970)
  970 FORMAT(54HOTHE PLOTTING CHARACTERS FOR THIS OBSERVATION ARE.... )
      DO 972 I=1,NPC
      L=PC(I)
  972 WRITE(6,971)I,L
  971 FORMAT(4H PC(I2,2H)=I1)
      GO TO 795
C ORDER THE VARIABLES
   13 DO 200 ISKIP=1,NVAR
      IP=IVAR(ISKIP)
  200 VV(IP)=V(ISKIP)
      IF(ICOUNT.NE.1)GO TO 795
      WRITE(6,7999)
 7999 FORMAT(48HOTHE VARIABLES IN THE FIRST OBSERVATION ARE...  )
      DO 650 I=1,NYS
  650 WRITE(6,651)I,VV(I)
  651 FORMAT(3H Y(I2,2H)=,E14.8)
      DO 652 I=1,NXS
      IXS=I+NYS
  652 WRITE(6,653)I,VV(IXS)
  653 FORMAT(3H X(I2,2H)=,E14.8)
      IF(DEM.EQ.THREED)GO TO 10013
C STORE THE VARIABLES
  795 KK=0
      DO 410 K=1,NVAR
      KSUB=((K-1)*NOBS)+KK+ICOUNT
      KMIN=KSUB+NOBS-ICOUNT+1
      KMAX=KMIN+1
      XMAT(KSUB)=VV(K)
      KK=KK+2
C STORE THE MAXIMUM AND MINIMUM VALUES
      IF(ICOUNT.EQ.1)GO TO 107
      AMAX=XMAT(KMAX)
      AMIN=XMAT(KMIN)
  120 IF((AMAX-VV(K))*(VV(K)-AMIN))106,106,410
  106 IF(VV(K)-AMAX)108,107,107
  107 AMAX=VV(K)
      XMAT(KMAX)=AMAX
```

CONTINUED

```
      IF(ICOUNT.NE.1)GO TO 410
  108 AMIN=VV(K)
      XMAT(KMIN)=AMIN
  410 CONTINUE
C STORE THE PLOTTING CHARACTERS
      IF(DEM.NE.THREED)GO TO 500
      DO 401 L=1,NPC
      KSLOT=(NOBS*2)*(NVAR+L-1)+ICOUNT
  401 XMAT(KSLOT)=PC(L)
  500 IF(ICOUNT.NE.NOBS)GO TO 103
C
C
C THE VALUES IN XMAT ARE CHANGED TO ROW AND COLUMN NUMBERS FOR THE
C          OUTPUT MATRIX(IGRAPH)
C
C
      KX=1
      DO 1050 K=1,NVAR
      NL=(K-1)*NOBS+KX
      ML=(NOBS*K)+KX-1
      KX=KX+2
      DO 1050 LK=NL,ML
      IF(K.GT.NYS)GO TO 1010
      DIV=50.
      GO TO 1011
 1010 DIV=100.
 1011 XINC=(XMAT(ML+2)-XMAT(ML+1))/DIV
      IF(XINC.EQ.0.)XINC=1.0
      A=XMAT(ML+1)
      RXMAT=0.
 1007 IF(A.GT.XMAT(LK))GO TO 1006
      A=A+XINC
      RXMAT=RXMAT+1.
      GO TO 1007
 1006 XMAT(LK)=RXMAT
      IF(K.GT.NYS)GO TO 1020
      IF(XMAT(LK).LE.50.)GO TO 1050
      XMAT(LK)=50.
      GO TO 1050
 1020 IF(XMAT(LK).LE.100.)GO TO 1050
      XMAT(LK)=100.
      GO TO 1050
  800 WRITE(6,990)NYS
      GO TO 1000
  801 WRITE(6,991)NXS
      GO TO 1000
  802 WRITE(6,992)NPC
      GO TO 1000
  803 WRITE(6,993)NPC
      GO TO 1000
  810 WRITE(6,994)NOBS
      GO TO 1000
  830 WRITE(6,995)
      GO TO 1000
  835 WRITE(6,996)ICOUNT,I
      GO TO 1000
  860 NOVER=((NYS+NXS)*NOBS)
      WRITE(6,997)NOVER
      GO TO 1000
  861 NOVER=((NYS+NXS+NPC)*NOBS)

   *
```

CONTINUED

```
      WRITE(6,998)NOVER
      GO TO 1000
C THE FOLLOWING CALL STATEMENT IS USED TO BACKSPACE THE INPUT UNIT
 999  CALL IOEDIT(1)
      READ(5,892)(R(I),I=1,20)
      WRITE(6,891)
      WRITE(6,893)(R(I),I=1,20)
      GO TO 1000
 990  FORMAT(5HONYS=I2,/44H NYS MUST BE GREATER THAN 0 AND LESS THAN 50)
 991  FORMAT(5HONXS=I2,/44H NXS MUST BE GREATER THAN 0 AND LESS THAN 50)
 992  FORMAT(5HONPC=I2,/44H NPC MUST BE BLANK OR 0                      )
 993  FORMAT(5HONPC=I2,/44H NPC MUST BE GREATER THAN 0 AND LESS THAN 11)
 994  FORMAT(6HONOBS=I3,/46H NOBS MUST BE GREATER THAN 1 AND LESS THAN 5
     101)
 995  FORMAT(63HOTHERE IS A VARIABLE LABEL WITH THE 3D PLOTTING CHARACTE
     1R GROUP)
 996  FORMAT(49HOTHERE IS A NON-NUMERIC CHARACTER IN OBSERVATION I3,24H
     1FOR PLOTTING CHARACTER I2)
 997  FORMAT(18HO(NYS+NXS) X NOBS=I6,/42H THIS MUST BE EQUAL TO OR LESS
     1THAN 10,000)
 998  FORMAT(22HO(NYS+NXS+NPC) X NOBS=I6,/42H THIS MUST BE EQUAL TO OR L
     1ESS THAN 10,000)
 891  FORMAT(60HOTHE FOLLOWING CONTROL CARD IS MISPUNCHED OR OUT OF ORDE
     1R...)
 892  FORMAT(20A4)
 893  FORMAT(1H 20A4)
1000 WRITE(6,1001)
1001 FORMAT(27HOUNABLE TO PROCESS THIS JOB/55H THE PROGRAM IS NOW SEARC
     1HING FOR THE NEXT JOB(IF ANY)  )
1002 READ(5,1003)TAL
1003 FORMAT(A4)
      DATA EE,CC/4HDONE,4HCONT/
      IF(TAL.EQ.EE)GO TO 2000
      IF(TAL.EQ.CC)GO TO 799
      GO TO 1002
1050 CONTINUE
C
C
C CONSTRUCT THE GRAPHS (IGRAPH)
C
C
      IF(DEM.EQ.TWOD)NPC=1
      DO 2010 J=1,NPC
      DO 2010 K=1,NYS
      KX=(2*(K-1))+1
      KY=(2*NYS)+1
      LNYS=NYS+1
      DO 2010 L=LNYS,NVAR
      REWIND 9
      DO 2030 I=1,50
      DO 2030 M=1,100
2030  IGRAPH(I,M)=100
      MT=(K-1)*NOBS+KX
      NK=(L-1)*NOBS+KY
      MK=NK+NOBS-1
      KY=KY+2
      ICNT=0
      ID=0
      DO 2011 LK=NK,MK
      IY=XMAT(MT)+.001
```

CONTINUED

```
      IX=XMAT(LK)+.001
      IF(DEM.EQ.THREED)GO TO 2015
      IF(IGRAPH(IY,IX).EQ.110)GO TO 3000
      IGRAPH(IY,IX)=IGRAPH(IY,IX)+1
      GO TO 2011
 3000 WRITE(9,3002)IY,IX
 3002 FORMAT(2I3)
      ID=ID+1
      GO TO 2011
 2015 ICNT=ICNT+1
      KSLOT=(NOBS+2)*(NVAR+J-1)+ICNT
      KPLOT=XMAT(KSLOT)+.001
      IF(IGRAPH(IY,IX).NE.100)GO TO 3001
      IGRAPH(IY,IX)=KPLOT
      GO TO 2011
 3001 WRITE(9,3003)IY,IX,KPLOT
 3003 FORMAT(3I3)
      ID=ID+1
 2011 MT=MT+1
      END FILE 9
      REWIND 9
C PRINT A GRAPH
      WRITE(6,3041)IJOB
 3041 FORMAT(1H1,15X,11HJOB NUMBER I3)
      WRITE(6,3040)(TITLE(I),I=1,15)
 3040 FORMAT(1H ,15X,15A4)
      IF(DEM.EQ.THREED)GO TO 3042
      WRITE(6,3043)
 3043 FORMAT(1H ,15X,23HTWO-DIMENSIONAL GRAPH     )
      GO TO 3044
 3042 WRITE(6,3045)J,(CONLAB(J,JC),JC=1,2)
 3045 FORMAT(1H ,15X,27HTHREE-DIMENSIONAL GRAPH,PC(I2,16H) IS DEFINED AS
     1 2A4)
 3044 CONTINUE
      DO 3050 IY=1,50
      DO 3050 IX=1,100
      IF(IGRAPH(IY,IX).NE.100)GO TO 3051
      IGRAPH(IY,IX)=IG(1)
      GO TO 3050
 3051 IF(IGRAPH(IY,IX).GT.100)IGRAPH(IY,IX)=IGRAPH(IY,IX)-100
      DO 3053 I=1,10
      IF(IGRAPH(IY,IX).NE.I)GO TO 3053
      IGRAPH(IY,IX)=IG(I+1)
      GO TO 3050
 3053 CONTINUE
      IGRAPH(IY,IX)=IG(11)
 3050 CONTINUE
      YINC=XMAT(MT+1)-XMAT(MT)
      YINC=YINC/50.
      BEGIN=XMAT(MT+1)-(.5*YINC)
      DO 3055 IJK=1,50
      I=50-IJK+1
      IF(I.EQ.50.OR.I.EQ.40.OR.I.EQ.30.OR.I.EQ.20.OR.I.EQ.10)GO TO 3060
      WRITE(6,3061)ILABEL(K,IJK),(IGRAPH(I,II),II=1,100)
 3061 FORMAT(1H A1,14X,1H.100A1)
      GO TO 3055
 3060 WRITE(6,3063)ILABEL(K,IJK),BEGIN,(IGRAPH(I,II),II=1,100)
 3063 FORMAT(1H A1,2X,E11.5,2H..100A1)
 3055 BEGIN=BEGIN-YINC
 3071 WRITE(6,3070)BEGIN
```

CONTINUED

```
3070 FORMAT(4H    E11.5,102H.....
     1.....)
      WRITE(6,3080)
3080 FORMAT(1H 15X,1H.19X,1H.19X,1H.19X,1H.19X,1H.19X,1H.)
      XINC=(XMAT(MK+2)-XMAT(MK+1))/100.
      XBEGIN=XMAT(MK+1)-(.5*XINC)
      DO 6070 III=20,100,20
      III=III/20
      XIK=III
6070  XABS(III)=XBEGIN+(XIK*XINC)
      WRITE(6,6071)XBEGIN,(XABS(I),I=1,5)
6071  FORMAT(1H 10X,E11.5,5(9X,E11.5))
      WRITE(6,3081)(ILABEL(L,II),II=1,50)
3081  FORMAT(1H015X,50A1)
      WRITE(6,3082)ID
3082  FORMAT(31H NUMBER OF ATTEMPTED OVERPLOTS=I3)
      IF(ID.EQ.0)GO TO 2010
      WRITE(6,3083)
3083  FORMAT(37H1COORDINATES OF ATTEMPTED OVERPLOTS      )
C PRINT THE OVERPLOTS
      DO 3085 I=1,ID
      IF(DEM.EQ.THREED)GO TO 3090
      READ(9,4040)IY,IX
4040  FORMAT(2I3)
      YI=IY
      YI=(BEGIN+(YI*YINC))
      XI=IX
      XI=(XBEGIN+(XI*XINC))
      WRITE(6,3091)YI,XI
3091  FORMAT(3H Y=E11.5,3X,2HX=E11.5)
      GO TO 3085
3090  READ(9,4041)IY,IX,KPLOT
4041  FORMAT(3I3)
      YI=IY
      YI=(BEGIN+(YI*YINC))
      XI=IX
      XI=(XBEGIN+(XI*XINC))
      WRITE(6,3092)YI,XI,KPLOT
3092  FORMAT(3H Y=E11.5,3X,2HX=E11.5,3X,3HPC=I2)
3085  CONTINUE
2010  CONTINUE
      READ(5,3020)TAIL,TAIM
      IF(TAIL.NE.EE.AND.TAIL.NE.CC)GO TO 999
3020  FORMAT(2A4)
      WRITE(6,3021)TAIL,TAIM
3021  FORMAT(1H12A4)
      IF(TAIL.EQ.CC)GO TO 799
2000  CONTINUE
      STOP
      END
```

Acknowledgment

Credit is due James A. Shaffer, computer programmer at the Ohio State University Computer Center, for the initial concepts of the program described in this paper.

The author also expresses gratitude to Chester E. Jensen, former Biometrics Section Chief at the Northeastern Forest Experiment Station, for technical review and suggestions for improvement during the course of developing the program.